Binary File Layout Managers

Wilf LaLonde and John Pugh

Introduction

There are many applications totally unrelated to Smalltalk that rely on binary data provided through files. If this data can be generated by a Smalltalk program, much care is needed to ensure that the data is output in exactly the right format. An example of such an application is a real-time 3D game engine that loads and plays “game world” files. Smalltalk itself is probably not a suitable vehicle for implementing such an engine but it can be used as a tool for building game worlds. Once a game world is designed in a Smalltalk environment, it is a manageable task to output it in a format suitable for the game engine.



Such a file can be viewed as a “game image”, not all that different from a “Smalltalk image”. Typically, such files contain data items like bytes, shorts, longs, strings, and, of course, pointers. The obvious way to output a pointer is to output a file offset where the first byte in the file is at offset 0, the second byte at offset 1, etc. When the file is read into memory and placed at address A, the reader (assuming it knows the exact format of the data) can replace each pointer offset O by “A+O”. This process is termed address relocation.



We had the opportunity to do something like this over a year ago. Initially, the amount of data that had to be output was small. So it wasn’t too difficult to compute the value of each pointer offset at the time it was output. However, over time, the amount of data and the number of different types of C structures that had to be handled grew much more than we had expected. In some cases, the pointers were backward references to data already output; in other cases, they were forward references. What we did was prespecify the number of bytes needed for each structure, precompute the number of bytes needed for each structure section (elements of the same type were generally output as consecutive elements) before output began, and also compute the starting offsets for each structure section. At output time, we used this information to compute what the pointer offsets had to be.



When we reached about 20 structures, the precomputation process itself was starting to get messy. Every object that we wanted to be able to output provided special methods to facilitate this process. Additionally, the amount of precomputed  data itself proved substantial. It wasn’t so much that it didn’t work but rather that it became error prone to change existing structures or add new ones that went between existing structures. There had to be a simpler way.

Layout Managers

We postulated that it might be possible to design a special kind of object for dealing with the pointer relocation problem and for managing the output of the C structures. We called it a layout manager. 



To develop this layout manager, we chose a new domain consisting of classes Client (Listing 1) and Phone (Listing 2). A client maintains a name and a collection of phones. A phone maintains the three component parts of a phone number: the area code (3 digits), the city code (3 digits), and the local number (4 digits). 



Our goal is to be able to generate a file with 4 sections: a header section, a clients section, a names section, and a phones section. Each section is described below:



	Header section

		pointer to first client (4 bytes)

		number of clients (4 bytes)

		pointer to first name (4 bytes)

		number of names (4 bytes)

		pointer to first phone (4 bytes)

		number of phones (4 bytes)



	Clients section (0 or more instances)

		pointer to name (4 bytes)

		pointer to first phone (4 bytes)

		number of phones (4 bytes)



	Names section (0 or more instances)

		characters in name (0 or more bytes)

		character 0 (1 byte)



	Phones section (0 or more instances)

		integer for area code (2 bytes)

		integer for city code (2 bytes)

		integer for local (2 bytes)



To do this, we will first need to tell the layout manager how many sections we need by specifying their names as follows:



	aLayoutManager

		layout: #(header clients 

names phones).



Next, to output an object such as a phone, we simply execute:



	aLayoutManager

begin: #phones;

emitShort: 613;

emitShort: 555;

emitShort: 1212;

end



It is the layout manager’s job to output the data into the appropriate section. To make it easier for the user, we permit nested begin-end sections provided the nested sections are other sections. This nesting facility also makes forward references easy. Consider



aLayoutManager

begin: #clients;

emitReferenceToNext: #names;

begin: #names;

emitString: 'joe';

end;

end.



If we want a reference to the start of a section or the final size of a section, we simply execute



aLayoutManager

emitReference: #clients;

emitSize: #clients



Finally, after everything has been emitted, we need to ask the manager to package everything up into one file and output it.



aLayoutManager

dump: fileName



We created a collection of client objects to test our application in class method exampleClients (see Listing 3). See “exampleLayout1: fileName” for a complete example using the layout manager. As an alternative to begin-end methods, we also permit the following replacement:



	aLayoutManager for: #phones do: [

aLayoutManager

emitShort: 613;

emitShort: 555;

emitShort: 1212]



See “exampleLayout2: fileName” for a direct translation of  “exampleLayout1: fileName” to use the for:do: notation.

The LayoutSection Class

Before discussing the LayoutManager class, let’s first consider the supporting LayoutSection class (see Listing 4). This class, which is private to the layout manager, represents one section. It maintains a name, stream, element size (in bytes), element count, and streamSize (in bytes) along with a few other instance variables. It is responsible for managing a section which entails tracking each element of the section as it is being emitted. As such, it tracks whether or not each element is the same size and how many there are. This is done transparently whenever a matching pair of begin-end messages is sent.



	aLayoutSection begin

		…

	aLayoutSection end



Method begin records the stream position in an instance variable called beginPosition and method end uses the new stream position to compute the size of one element. Method end also determines whether or not the element size changes and counts the number of elements. Method hasFixedSizeElements can be used to determine whether or not the elements are all the same size.



After all data has been emitted, a layout section is provided with its own offset into the final file. Given an offset into a section, this can be used to compute an offset in the final file.

The LayoutManager Class

The LayoutManager class was designed to maintain four pieces of information: (1) sectionNames (symbols for the section names indicating the order in which the sections are to be emitted), (2) sections (a dictionary of layout sections indexed by the section names),  (3) activeSections  (a stack of layout sections for which a “begin” has been issued but not the corresponding “end”), and (4) fixups (a collection of messages that when executed at the end will compute the unresolved values).



The core of the layout manager is used by (1) beginning a section to make it the active section, (2) emitting any number of data items, and then finally (3) ending the section. Items are always emitted into the active section by simply appending the information to the section’s associated stream. Fixups (in the current design) are needed only when the following emit messages are used:



aLayoutManager

emitReferenceToNext: sectionName;

emitReference: sectionName;

emitSize: sectionName



The first two emit messages are similar; emitReferenceToNext: references the section’s current stream position while emitReference: reference the start of the stream. They are both implemented using method “emitReferenceTo: sectionName at: sectionOffset” which in turn uses “emitFixSectionReferenceTo: sectionName at: sectionOffset” and then emits an integer 0 using emitLong:, a 32 bit integer that will be replaced by the correct value after everything has been emitted.



Message emitSize: is implemented similarly; it uses “emitFixSectionSize: sectionName” and then emits an integer 0 to be later fixed up.



In order to be able to fix the integers (currently emitted as zeros), we construct message objects. Each message object can be constructed as follows:



    Message

        receiver: anObject

        selector: aSymbol

        arguments: (Array

            with: argument1

            with: argument2

					…

            with: argumentn)



Sending the message “perform” to the message object causes it to execute. For example,



| aMessage |

aMessage := Message

receiver: 'hi'

selector: #at:put:

arguments: (Array

with: 2

with: $o).

	aMessage perform



is equivalent to executing expression “'hi' at: 2 put: $0”.



To conclude, our method “emitFixSectionReferenceTo: otherSectionName at: otherSectionOffset” constructs a message object that when executed looks like the following:

 

		Fixup message object for Section Reference

		currentSection

			at: currentSectionOffset

			putRelocatedOffset: otherSectionOffset

			in: otherSection



Similarly, method “emitFixSectionSize: sectionName” construct a message object that looks like the following:

 

		Fixup message object for Section Size

		currentSection

			at: currentSectionOffset

			putSizeOf: otherSection



When these message are ultimately executed (in method dump:), each section knows its offset in the overall file and its element count. The section can easily reposition its stream at “currentSectionOffset” to replace the previously emitted zero by the corrected value. See the LayoutSection class in Listing 2 for the actual methods. 

Conclusions

We showed how we might design a layout manager for simplifying the task of emitting binary data for applications written in other languages. As currently designed, the layout manager is very easy to use provided we don’t need multiple references to the same locations. We also didn’t bother with backward references since we didn’t need it for the example we used. Such extensions don’t seem to be too difficult to add.





Listing 1 The Client class.



class:								Client

superclass:						Object

instance variables:		name phones



class methods



instance creation

new

    ^super new initialize



examples

example1

    "Client example1"

    ^Client new

        name: 'Tim';

        addPhone: (Phone from: '613-555-1212');

        addPhone: (Phone from: '613-555-4552');

        addPhone: (Phone from: '613-555-0022')



instance methods



initializing

initialize

    self name: 'Unknown'.

    phones := OrderedCollection new



get/set

name: aString

    name := aString

name

    ^name

phones

    ^phones



adding

addPhone: aPhone

    self phones add: aPhone



printing

printOn: aStream

    super printOn: aStream.

    aStream nextPutAll: ' called '.

    self name printOn: aStream



Listing 2 The Phone class.



class:								Phone

superclass:						Object

instance variables:		areaCode cityCode local



class methods



instance creation   

from: aString

    "Doesn't perform error checking."

    ^self new

        areaCode: (aString copyFrom: 1 to: 3) asInteger;

        cityCode: (aString copyFrom: 5 to: 7) asInteger;

        local: (aString copyFrom: 9 to: 12) asInteger



examples

example1

    "Phone example1"

    ^Phone from: '613-555-0012'



instance methods



get/set

areaCode

    ^areaCode

areaCode: aString

    areaCode := aString

cityCode

    ^cityCode

cityCode: aNumber

    cityCode := aNumber

local

    ^local

local: aNumber

    local := aNumber



printing

printOn: aStream

    aStream

        nextPutAll: '(';

        nextPutAll: (self pad: self areaCode to: 3);

        nextPutAll: ')-';

        nextPutAll: (self pad: self cityCode to: 3);

        nextPutAll: '-';

        nextPutAll: (self pad: self local to: 4)



pad: integer to: size

    | string |

    string := integer printString.

    string size >= size ifTrue: [^string].

    ^((String new: size - string size)

        atAllPut: $0), string



Listing 3 The LayoutManager class.



class:								LayoutManager

superclass:						Object

instance variables:		sectionNames sections activeSections 										fixups



class methods



instance creation

new

    ^super new initialize



examples

exampleClients

    "LayoutManager exampleClients"

    | manager clients client clientCount |

    clientCount := 0.

    clients := #(2 1 3) collect: [:phoneCount |

        clientCount := clientCount + 1.

        client := Client new

            name: 'client', clientCount printString.

        1 to: phoneCount do: [:phoneIndex |

            client addPhone: (Phone

                from: '613-555-000', phoneIndex printString)].

        client].

    ^clients



exampleLayout1: fileName

    "LayoutManager exampleLayout1: 'c:\poof\junk'"

    | manager |

    manager := LayoutManager new.

    manager

        layout: #(header clients names phones);

        begin: #header;

            emitReference: #clients;

            emitSize: #clients;

            emitReference: #names;

            emitSize: #names;

            emitReference: #phones;

            emitSize: #phones;

        end.

    self exampleClients do: [:client |

        manager

            begin: #clients;

                emitReferenceToNext: #names;

                emitReferenceToNext: #phones;

                emitLong: client phones size;

                begin: #names;

                    emitString: client name;

                end.

                client phones do: [:phone |

                    manager

                        begin: #phones;

                            emitShort: phone areaCode;

                            emitShort: phone cityCode;

                            emitShort: phone local;

                        end].

        manager

            end].

    manager dump: fileName



exampleLayout2: fileName

    "LayoutManager exampleLayout2: 'c:\poof\junk'"

    | manager |

    manager := LayoutManager new

        layout: #(header clients names phones).

    manager for: #header do: [

        manager

            emitReference: #clients;

            emitSize: #clients;

            emitReference: #names;

            emitSize: #names;

            emitReference: #phones;

            emitSize: #phones].

    self exampleClients do: [:client |

        manager for: #clients do: [

            manager

                emitReferenceToNext: #names;

                emitReferenceToNext: #phones;

                emitLong: client phones size.

            manager for: #names do: [

                manager emitString: client name].

            client phones do: [:phone |

                manager for: #phones do: [

                    manager

                        emitShort: phone areaCode;

                        emitShort: phone cityCode;

                        emitShort: phone local]]]].

    manager dump: fileName



instance methods



initializing

initialize

    activeSections := OrderedCollection new.

    fixups := OrderedCollection new.



layout/grouping

layout: anArray

    sectionNames := anArray.

    sections := IdentityDictionary new.

    sectionNames do: [:name |

        sections

            at: name

            put: (LayoutSection new name: name)].

 

for: sectionName do: aBlock

    self begin: sectionName.

    aBlock value.

    self end



layout/grouping support

begin: sectionName

    activeSections add: (sections at: sectionName).

    self activeSection begin

end

    self activeSection end.

    activeSections removeLast.



emitting   

emitByte: anInteger

    self activeStream nextBytePut: anInteger

emitShort: anInteger

    self activeStream nextUShortPut: anInteger

emitLong: anInteger

    self activeStream nextULongPut: anInteger

emitString: aString

    "Output with \0 terminator."

    self activeStream

        putBytesFrom: aString;

        nextBytePut: 0

emitSize: sectionName

    self emitFixSectionSize: sectionName.

    self emitLong: 0

emitReference: sectionName

    self

        emitReferenceTo: sectionName

        at: 0

emitReferenceToNext: sectionName

    self

        emitReferenceTo: sectionName

        at: (sections at: sectionName) stream position



emitting support

activeSection

    ^activeSections last

activeStream

    ^self activeSection stream

activePosition

    ^self activeStream position

emitReferenceTo: sectionName at: sectionOffset

    self 

		emitFixSectionReferenceTo: sectionName 

		at: sectionOffset.

    self emitLong: 0



fixups

emitFixSectionReferenceTo: sectionName at: sectionOffset

    | section |

    section := sections at: sectionName.

    fixups add: (self 

		referenceFixupMessageTo: section at: sectionOffset)



emitFixSectionSize: sectionName

    | section |

    section := sections at: sectionName.

    fixups add: (self sizeFixupMessageTo: section).



referenceFixupMessageTo: section at: sectionOffset

    ^Message

        receiver: self activeSection

        selector: #at:putRelocatedOffset:in:

        arguments: (Array

            with: self activePosition

            with: sectionOffset

            with: section)



sizeFixupMessageTo: section

    ^Message

        receiver: self activeSection

        selector: #at:putSizeOf:

        arguments: (Array

            with: self activePosition

            with: section)



dumping

dump: fileName

    self

        computeSectionOffsets;

        executeFixups;

        appendSectionsInto: fileName



dumping support

computeSectionOffsets

    "Compute offset for each section name; use the section order."

    sections inject: 0 into: [:offset :section |

        section offset: offset.

        offset + section streamSize]



executeFixups

     fixups do: [:message | message perform]



appendSectionsInto: fileName

    | stream section |

    stream := File pathName: fileName.

    sectionNames do: [:name |

        section := sections at: name.

        stream nextPutAll: section contents].

    stream close



Listing 4 The LayoutSection class.



class:								LayoutSection

superclass:						Object

instance variables:		name stream offset elementType 

										elementSize elementCount streamSize 

										beginPosition



class methods



instance creation  

new

    ^super new initialize



instance methods



initializing

initialize

    self

        name: '';

        stream: (WriteStream on: (String new: 1000));

        offset: 0.

    elementType := #unknownSize.

    elementSize := 0.

    elementCount := 0.

    streamSize := 0



get/set  

name

    ^name

name: aString

    name := aString

stream

    ^stream

stream: aStream

    stream := aStream

offset

    ^offset

offset: anInteger

    offset := anInteger

elementSize

    ^elementSize

elementCount

    ^elementCount

streamSize

    ^streamSize



grouping

begin

    beginPosition := self stream position



end

    self reviseElementSize.

    elementCount := elementCount + 1.

    streamSize := self stream position



reviseElementSize

    | endPosition size |

    elementType == #varyingSize ifTrue: [

        ^self].

    endPosition := self stream position.

    size := endPosition - beginPosition.

    elementType == #unknownSize ifTrue: [

        elementType := #fixedSize.

        elementSize := size.

        ^self].

    "Default: elementType == #fixedSize"

    self elementSize = size ifTrue: [^self].

    elementType := #varyingSize



queries

hasFixedSizeElements

    ^elementType ~~ #varyingSize

contents

    ^self stream

        position: self streamSize;

        contents



fixup choices

at: position putRelocatedOffset: amount in: section

    self stream

        position: position;

        nextULongPut: section offset + amount

   

at: position putSizeOf: section

    section hasFixedSizeElements ifFalse: [

        self error: 'section elements not fixed size'].

    self stream

        position: position;

        nextULongPut: section elementCount



printing 

printOn: aStream

    aStream nextPutAll: 'section '.

    self name printOn: aStream





� PAGE �1�








